ECE 683/685: Project Description

Comments

The scanner will process and discard all whitespace and comments. We will assume a C++ short comment style that start with the string "//" and continue to the next newline character. In addition to comments, whitespace is defined as the space character, newline characters, and tab characters. Terminals are in bold font and non-terminals are placed in angled brackets "<>"; be careful not to confuse BNF operators with terminals. If unclear, ask. The start state for this grammar is the very first grammar rule below.

Syntax

<program> ::=
  <program_header> <program_body>

<program_header> ::= **program** <identifier> **is**

<program_body> ::=
    (<declaration> **;**)*
  begin
    (<statement> **;**)*
  end program

<declaration> ::=
    <procedure_declaration>
| <variable_declaration>

<procedure_declaration> ::=
  <procedure_header> <procedure_body>

<procedure_header> :: =
  procedure <identifier>
    ([<parameter_list>] **)**

<parameter_list> ::=
  <parameter> **,** <parameter_list>
| <parameter>

<parameter> ::= <variable_declaration> (**in** | **out**)

<procedure_body> ::=
    (<declaration> **;**)*
  begin
    (<statement> **;**)*
  end procedure

<declaration> ::=
  [**global**] <procedure_declaration>
| <variable_declaration>

<variable_declaration> ::=
  [**global**] <type_mark> <identifier>
  [**[** <array_size> **]**]

<type_mark> ::=
  integer
| **float**
| **bool**
| **string**

<array_size> ::= <number>

<statement> ::=
    <assignment_statement>
| <if_statement>
| <loop_statement>
| <return_statement>

<procedure_call> ::=
    <identifier> **(** [<argument_list>] **)**

<assignment_statement> ::=
    <destination> := <expression>

<destination> ::=
    <identifier> [**[** <expression> **]**]

<if_statement> ::=
    if <expression> **then** (<statement> **;**)+
    [**else** (<statement> **;**)+]
    end if

<loop_statement> ::=
    for <assignment_stmt> <expression>
        (<statement> **;**)*
    end for

<return_statement> ::= **return**

<identifier> ::= [**a**-**z**A-**Z**][**a**-**z**A-**Z0**-**9**_]*

<expression> ::=
    <expression> **&** <arithOp>
| <expression> | <arithOp>
| [**not**] <arithOp>

<arithOp> ::=
    <arithOp> + <relation>
| <arithOp> - <relation>
| <relation>

<relation> ::=
    <relation> < <term>
| <relation> >= <term>
| <relation> <= <term>
| <relation> > <term>
| <relation> == <term>
| <relation> != <term>
| <term>

<term> ::=
    <term> * <factor>
| <term> / <factor>
| <factor>

<factor> ::=
    (<expression> **)**
| <function_call>
| [-] <name>
| [-] <number>
| <string>
| **true**
| **false**

<name> ::=

<identifier> [**[** <expression> **]**]

<argument_list> ::=

<expression> **,** <argument_list>

| <expression>

<number> ::= [**0-9**][**0-9_**]*[**.**[**0-9_**]*]

<string> :: = **"**[**a-zA-Z0-9 _,;:.'**]***"**

1. Semantics

1. Procedure parameters are transmitted by value. Recursion is supported.
2. Non-local variables and functions are not visible except for those variables and functions in the outermost scope prefixed with the **global** reserved word. Functions currently being defined are visible in the statement set of the function itself (so that recursive calls a possible).
3. No forward references are permitted or supported.
4. Expressions are strongly typed and types must match. However there is automatic conversion in the arithmetic operators to allow any mixing between integers and floats. Furthermore, the relational operators can compare booleans with integers (booleans are converted to integers as false → 0, true → 1).
5. The type signatures of a procedures arguments must match exactly their parameter declaration.
6. Arithmetic operations (add, sub, multiply, divide, bitwise and "&", bitwise or "|") are defined for integers and floats only. The bitwise and "&", bitwise or "|", and bitwise **not** operators are valid only on variables of type integer.
7. Relational operations are defined for integers and booleans. Only comparisons between the compatible types is possible . Relational operators return a boolean result.
8. In general, you should treat string variables as pointers to a null terminated string that is stored in your memory space. Strings are then read/written to/from this space using the get/put functions. The in memory string values cannot be destroyed or changed (although string variables can be assigned to point to different strings).
9. Parameter transmission is by value for integers and booleans; strings are passed by sending the memory pointer to the string. Output parameters are transmitted by value-result.

2. Final Report

In addition to sending me your compiler source and build environment that I can run on my linux workstation (you are responsible for ensuring that it will build on a standard linux box; if you build it on some exotic system like Haiku, we can discuss a demo on that platform), you must also turn in a one page report documenting your compiler. It should document your software, its structure, the build process, and highlight any unique features you have implemented in the system.

3. Change Log

Revision 0: 1/29/13

- Initial version.